


L'ouverture d'un fichier ouvre un classeur composé de feuilles de calculs.
La gestion des feuilles de calcul se fait via les onglets en bas de la feuille.



Faire un clic droit sur l'onglet pour ouvrir la liste des commandes : insertion de feuille / renommer la feuille...

Dans la feuille les cellules sont repérées par leurs coordonnées

Exemple

A13

Une plage de cellules a pour coordonnées :

A13:A25 où le séparateur est « : »

Une plage peut être discontinue, on utilise alors la combinaison des « : » et du « ; »

A13:A20;B10:B16

On sélectionne des cellules non contiguës avec l'appui sur la touche **CTRL** lors de la sélection. Très utile pour la mise en forme de la feuille : bordures, fonds...

Contenu des cellules

Les cellules peuvent contenir :

1. Des **nombres** : Utiliser le pavé numérique pour entrer les nombres, le séparateur décimal est la virgule. Les nombres entrés se calent par défaut à droite dans la cellule.
2. Du **texte** : Suite de caractères alphabétiques et de chiffres mélangés. Les textes se calent à gauche par défaut dans la cellule.
3. Une **formule de calcul** : Toute formule de calcul commence par le signe « = »

Les formules de calcul

On utilise dans les formules soit des fonctions soit les opérateurs arithmétiques courants (+, -, *, /)

Exemple

=(A15*1,196)/10

Concernant les fonctions de traitement, la syntaxe est toujours la même :

Le signe « = »

Le nom de la fonction suivi de parenthèses.

Les arguments dans les parenthèses séparés par des « ; »

=SOMME(A10;C17;D12)

Le seul cas particulier est l'usage d'une plage de cellules dans les parenthèses :

=MOYENNE(A13:A25)

À noter que la syntaxe est respectée car le seul argument de la fonction est ici la plage. S'il y a un autre argument ou retrouve le séparateur « ; »

=MOYENNE(A13:A25;B12)

Imbrication des fonctions

Les fonctions et calculs peuvent être imbriqués dans la formule. Dans ce cas, on respecte les règles de syntaxe précédentes.

Exemple

Les notes de l'étudiant sont saisies dans les cellules A13, B13, C13 et D13. Si la moyenne des notes est supérieure ou égale à 10, l'étudiant est Reçu sinon Ajourné.

=SI(MOYENNE(A13:D13)>=10;"Reçu";"Ajourné")

	A	B	C	D	E
11					
12	Note 1	Note 2	Note 3	Note 4	Résultat
13	12	10	15	9	Reçu
14	6	11	10	5	Ajourné

Chaînes de caractères

À l'intérieur des formules, les chaînes de caractères sont saisies entre guillemets dans les cellules.

Pour faire afficher le texte **Reçu** ou le texte **Ajourné** dans la cellule dans l'exemple précédent, on place les chaînes entre guillemets.

À noter qu'il est possible d'utiliser la chaîne vide "" (sans espace entre les guillemets) pour que l'affichage soit vierge et la cellule reste vide.

On peut concaténer (accoler) des chaînes de caractères avec le signe &

Exemple

Si la cellule A10 contient le texte Paul, la cellule B10 contient la chaîne DURANT, en C10 :

=A10&" "&B10

fera afficher Paul DURANT

Le résultat obtenu :

	A	B	C	D	E
9	Alain	DUPOND	Alain DUPOND		
10	Paul	DURANT	Paul DURANT		
11					
12					

On concatène ici le prénom puis un espace puis le nom.

On pourra ainsi inclure un texte et une référence dans une cellule :

En D10 : ="Le nom du gagnant est "&C10

On pourra inclure un texte et un calcul dans une cellule :

Si les notes de Paul sont dans la plage G10:J10

= "Moyenne de "&A10&" : "&MOYENNE(G10:J10) fera afficher Moyenne de Paul : 11,5

A vous d'essayer.



Références relatives et absolues

Par défaut les références sont en mode relatif dans le tableur. Cela permet la recopie des formules mais pose problème dans certains cas où il faut faire référence à une cellule particulière.

On utilise alors **le signe \$ avant la référence** pour bloquer cette référence.

\$A10 : lors de la copie, la référence de colonne sera bloquée

B\$9 : lors de la copie, la référence de ligne sera bloquée

\$D\$21 : lors de la copie, la référence de colonne et de ligne seront bloquées. La référence de cellule reste bloquée. Ce cas sert pour faire toujours référence au contenu d'une cellule particulière lors de la copie.

Exemple

La cellule G1 contient le taux de TVA. Lors des calculs on utilise la référence \$G\$1.

Le résultat obtenu :

	A	B	C	D	E	F	G
1						Taux TVA	19,6%
2		Nombre	Prix article	Prix HT	Prix TTC		
3	Crayon	100	0,60 €	60,00 €	11,76 €		
4	Gomme	50	0,25 €	12,50 €	2,45 €		
5	Cahier	250	1,12 €	280,00 €	54,88 €		

Tests

On réalise un test lorsque le calcul dépend de conditions. On réalise ainsi le traitement/calcul selon les conditions requises.

Le test booléen rend des valeurs logiques VRAI ou FAUX comme réponse.

La fonction utilisée est **SI** avec la syntaxe :

SI(test;valeur_si_VRAI;valeur_si_FAUX)

Dans la fonction **SI** on réalise d'abord le test. Alors si la valeur renvoyée est 'VRAI' on réalise le traitement de **valeur_si_VRAI** sinon on réalise le traitement pour la valeur 'FAUX' soit **valeur_si_faix**

A noter que si le test rend la valeur 'VRAI', le premier bloc **valeur_si_VRAI** est traité, le traitement s'arrête et la valeur pour 'FAUX' n'est pas traitée.

Le test

Le bloc de test dans le **SI** comporte 3 éléments :

1. On teste sur le contenu d'une cellule. Donc le premier terme du test est une référence à une cellule.
2. Le second élément du test est un opérateur logique. *On teste si la valeur contenue dans la cellule de référence est égal, supérieur...*

Les opérateurs de test logique sont :

=	égal
>	supérieur
<	inférieur
>=	Supérieur ou égal
<=	Inférieur ou égal

3. La troisième partie du test est la valeur de comparaison.
- 4.

<i>Cela peut être :</i>	<i>Exemple</i>
La référence d'une cellule contenant cette valeur	On teste si le contenu de A10 est supérieur au contenu de D5 A10>D5
Une valeur sous forme d'un nombre	On teste si l'âge d'un candidat est supérieur à 18 ans A10>18

Une chaîne de caractères	On teste si la cellule contient le texte "grossiste" A10="grossiste"
Une formule de calcul faisant référence à 1 ou plusieurs cellules	On teste si le contenu de A10 est inférieur à la moyenne des notes contenues dans la plage B10:F10 A10<MOYENNE(B10:F10)

Exemple :

La cotisation est de 200€ si le type d'abonnement est « individuel », elle est de 350 € dans les autres cas.

On a donc en E3

On teste si le contenu de B3 est le texte "individuel" alors on prend la valeur 200 sinon, la valeur 350

=SI(B3="individuel";200;350)

	A	B	C	D	E	F	G
1							
2	Noms	Abonnement	Année inscription	Ancienneté	Cotisation (en Euros)	Réduction (en Euros)	Prix final (en Euros)
3	H	individuel	2008	3	200	0	200
4	F	couple	2007	4	350	52,5	297,5
5	G	couple	2006	5	350	52,5	297,5
6	D	individuel	2005	6	200	0	200
7	K	couple	2004	7	350	52,5	297,5

Pour les tests, il est utile d'écrire en texte français le traitement à réaliser en utilisant la forme :
Si le texte du test à réaliser alors je fais le traitement si vrai sinon je fais le traitement si faux
 Cette forme textuelle est facile à traduire ensuite dans la syntaxe du tableur.

Fonctions ET / OU

Ces fonctions logiques permettent de combiner des conditions dans les tests.

Syntaxe :

OU(condition1;condition2;condition3)

ET(condition1;condition2;condition3)

Comme pour toute fonction, parenthèses et points virgules pour séparer les conditions.

Exemple

Dans la suite de l'exemple précédent, on décide d'attribuer une réduction de 15% si l'abonnement est un « couple » **et** à partir de 4 ans d'ancienneté d'inscription. On a donc la formule suivante en F3 :

=SI(ET(D3>=4;B3="couple");E3*0,15;0)

	A	B	C	D	E	F	G
1							
2	Noms	Abonnement	Année inscription	Ancienneté	Cotisation (en Euros)	Réduction (en Euros)	Prix final (en Euros)
3	H	individuel	2008	3	200	0	200
4	F	couple	2007	4	350	52,5	297,5
5	G	couple	2006	5	350	52,5	297,5
6	D	individuel	2005	6	200	0	200
7	K	couple	2004	7	350	52,5	297,5
8	C	individuel	2003	8	200	0	200

Des fonctions à connaître

SOMME

Calcule la somme des valeurs contenues dans les cellules

SOMME(arg1;arg2;arg3;arg4) Ou SOMME(plage_de_cellules)

Ex: =SOMME(A3;B5;G17;K12)
=SOMME(B12:B53)

MOYENNE

Renvoie la moyenne des arguments. La fonction MOYENNE ignore les cellules contenant du texte ou les cellules vides.

Ex: =MOYENNE(B5:G5)

MAX

Donne la valeur la plus grande parmi une liste de valeurs. Ignore les valeurs logiques et le texte.

Ex: =MAX(B15:B30)

MIN

Donne la valeur la plus petite parmi une liste de valeurs. Ignore les valeurs logiques et le texte.

Ex: =MIN(B15:B30)

NB

Détermine le nombre de cellules d'une plage qui contiennent des nombres.

Ex: =NB(A5:A40)

NB.SI

Détermine le nombre de cellules non vides répondant à la condition à l'intérieur d'une plage.

Ex:

=NB.SI(A2:A15;"pommes")	Nombre de cellules contenant « pommes » dans la plage A2 à A15.
=NB.SI(B2:B15;">55")	Nombre de cellules dont la valeur est supérieure à 55 dans la plage B2 à B15.
=NB.SI(A2:A15;A3)	La cellule A3 contenant le mot « Pêches » La formule renvoie le nombre de cellules contenant « Pêches » dans la plage A2 à A15.

NB.VIDE

Compte le nombre de cellules vides à l'intérieur d'une plage.

Ex: =NB.VIDE(A10:A45)

NB.VAL

Détermine le nombre de cellules non vides dans la plage spécifiée.

Ex: =NB.VAL(A2:A8)